Manual Solution for the transport Problem

(% i1) ratprint: false$

1 Preparations



Figure 1: Basic meshing of a 1D domain (Using a FV Mesh)

Figure 1: The Mesh we'll work on

How the gradient of ϕ at the face should be calculated (between cells i and j)?

(% i2) grad(\phi,i,j,dx) := (\phi[i]-\phi[j])/dx;

(% o2) $$\operatorname{grad}(\phi, i, j, dx) := \frac{\phi_i - \phi_j}{dx}$$

How to interpolate ϕ values from cell centers (between cells i and j) to face centers?

(% i3) face_inter(\phi,i,j):= (\phi[i]+\phi[j])/2;

(% o3) $$\operatorname{face\_inter}(\phi, i, j) := \frac{\phi_i + \phi_j}{2}$$

Gauss-Seidel iterations

(% i4)
```
gauss_seidel(A,b,iters):=block(
    n:rank(A),
    x:float(zeromatrix(n,1)),
    y:float(zeromatrix(n,1)),
    dif:1,
    for m:1 thru iters do(
        for i:1 thru n do(
            y[i,1]:(b[i,1]-sum(A[i,j]*y[j,1],j,1,i-1)-sum(A[i,j]*x[j,1],j,i+1,n))/A[i,i]
        ),
        dif:sort (makelist(abs(x[i,1]-y[i,1]),i,1,n), 'ordergreatp),
        print ("Iteration ",m, ": ", transpose(y), ",dif=", dif[1]),
        x:copymatrix(y)
), return(x))$
```

Convective and diffusive setup

(% i7)
```
U:0.03$ K:0.01$ dx:0.1$
```

2 Algebraic Equations for cells

For all internal cells (i going from 1 to 8):

(% i9)
```
X:[\phi[i-1], \phi[i], \phi[i+1]]$ icell: U*face_inter(X,3,2)-K*grad(X, 3,2,dx) - U*face_inter(X,2,1) +K*grad(X, 2,1,0.1),ratsimp;
```

(icell)
$$-\frac{17\phi_{i+1}-40\phi_i+23\phi_{i-1}}{200}$$

(% i10)
```
icell, expand, float;
```

(% o10)
$$-0.085\phi_{i+1}+0.2\phi_i-0.115\phi_{i-1}$$

(% i13)
```
X:[\phi[0], \phi[1]]$ xb:1$ cell0: U*face_inter(X,1,2)-K*grad(X, 2,1,0.1) - U*xb - K*(xb-X[1])/(dx/2), expand, float;
```

(cell0)
$$-0.085\phi_1+0.315\phi_0-0.23$$

(% i16)
```
X:[\phi[8], \phi[9]]$ xb:0$ cell9: -U*face_inter(X,1,2)+K*grad(X, 2,1,0.1) + U*xb - K*(xb-X[2])/(dx/2), expand, float;
```

(cell9)
$$0.285\phi_9-0.115\phi_8$$

3 Matrix Resolution

Collect coefficients from cells equations:

(%i17) A:matrix([0.315 , -0.085, 0, 0, 0, 0, 0, 0, 0], [-0.115, 0.2, -0.085, 0, 0, 0, 0, 0, 0], [0, -0.115, 0.2, -0.085, 0, 0, 0, 0, 0], [0, 0, -0.115, 0.2, -0.085, 0, 0, 0, 0], [0, 0, 0, -0.115, 0.2, -0.085, 0, 0, 0], [0, 0, 0, 0, -0.115, 0.2, -0.085, 0, 0], [0, 0, 0, 0, 0, -0.115, 0.2, -0.085, 0], [0, 0, 0, 0, 0, 0, -0.115, 0.2, -0.085], [0, 0, 0, 0, 0, 0, 0, -0.115, 0.285]);

(A)

$$\begin{pmatrix} 0.315 & -0.085 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -0.115 & 0.2 & -0.085 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -0.115 & 0.2 & -0.085 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -0.115 & 0.2 & -0.085 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -0.115 & 0.2 & -0.085 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -0.115 & 0.2 & -0.085 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -0.115 & 0.2 & -0.085 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -0.115 & 0.2 & -0.085 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -0.115 & 0.285 \end{pmatrix}$$

Source term present only in first cell

(%i18) B:matrix([0.23],[0],[0],[0],[0],[0],[0],[0],[0]);

(B)

$$\begin{pmatrix} 0.23 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{pmatrix}$$

Solve the system with Gauss-Seidel Method (run for 40 iterations and watch max solution change in an iteration)

(%i19) iterative_sol:gauss_seidel(A,B,40);

... Goes on about iteration results ...

(iterative_ sol)
$$\begin{pmatrix} 0.989130803064315 \\ 0.9598697431668418 \\ 0.9205309580963118 \\ 0.8676405533439283 \\ 0.7964673452312362 \\ 0.7005668030132802 \\ 0.5711678657073875 \\ 0.3963489778943668 \\ 0.1599302893257971 \end{pmatrix}$$

4 Solution from Theory

Forget some variable so we can continue using the same names

(% i21) kill(U, K, x, y)$ assume(U> 0);

(% o21) [U¿ 0]

Compose and solve the ODE:

(% i22) ode: U*'diff(y, x) - K*'diff(y,x,2) = 0;

(ode)
$$U \left(\frac{d}{dx} y \right) - K \left(\frac{d^2}{dx^2} y \right) = 0$$

(% i23) gsol: ode2(ode, y, x);

(gsol)
$$y = \%k1\, \%e^{\frac{Ux}{K}} + \%k2$$

Figure out k1 and k2 coeffs based on boundary values (involves solving a system of two equations)

(% i25) f0: rhs(ev(gsol, x=0))$ f9: rhs(ev(gsol, x=0.9))$

(% i26) sol:solve([f0=1, f9=0], [%k1, %k2]);

(sol)
$$[[\%k1 = -\frac{1}{\%e^{\frac{0.9U}{K}} - 1}, \%k2 = \frac{\%e^{\frac{0.9U}{K}}}{\%e^{\frac{0.9U}{K}} - 1}]]$$

(% i27) coeffs:ev(sol, U=0.03, K=0.01);

(coeffs) $[[\%k1 = -0.07204750205711648, \%k2 = 1.072047502057116]]$

(% i28) theory_sol:makelist(ev(rhs(gsol), x=i/10.0+0.05, U=0.03, K=0.01, %k1=rhs(coeffs[1][1]), %k2=rhs(coeffs[1][2])), i, 0, 8);

(theory_ sol)
$[0.9883402470641222, 0.9590545266468127, 0.9195229390052993, 0.8661608772499396, 0.7941296281990526, 0.69$

(% i29) iterative_sol-theory_sol;

(% o29) $$\begin{pmatrix} 7.90556000192843610^{-4} \\ 8.15216520029138410^{-4} \\ 0.001008019091012557 \\ 0.001479676093988691 \\ 0.00233771703218355 \\ 0.003669190766267927 \\ 0.005519746571607254 \\ 0.007869143024669012 \\ 0.01060262339579992 \end{pmatrix}$$